Journal of Software for Algebra and Geometry

The *SpecialFanoFourfolds* package in Macaulay2

GIOVANNI STAGLIANÒ

vol 14 2024

JSAG **14** (2024), 111–116
https://doi.org/10.2140/jsag.2024.14.111
The Journal of Software for
Algebra and Geometry

The *SpecialFanoFourfolds* package in Macaulay2

GIOVANNI STAGLIANÒ

ABSTRACT: We introduce the *Macaulay*2 package *SpecialFanoFourfolds*, a package that provides functions for working with cubic fourfolds, Gushel–Mukai fourfolds, and some other special Fano fourfolds.

INTRODUCTION. The package *SpecialFanoFourfolds* in *Macaulay*2 [2] provides support for some types of Hodge-special Fano fourfolds, which can be represented as smooth hypersurfaces of low degree r in some fixed ambient fivefold $\mathbb{V}$. Roughly speaking, a smooth degree-r hypersurface $X \subset \mathbb{V}$ is called Hodge-special if it contains an algebraic surface S whose cohomology class does not come from the ambient fivefold $\mathbb{V}$. In the parameter space of all smooth hypersurfaces $X \subset \mathbb{V}$ of degree r, the locus of Hodge-special fourfolds is called the Noether–Lefschetz locus.

One important case is that of smooth cubic hypersurfaces in $\mathbb{P}^5$ (cubic fourfolds for short). The Noether–Lefschetz locus in the 20-dimensional moduli space $\mathcal{C} = (\mathbb{P}(\mathcal{O}_{\mathbb{P}^5}(3)) \setminus \mathrm{Disc}^3_{\mathbb{P}^5})/\mathrm{PGL}_6$ of cubic fourfolds is a countable union of irreducible hypersurfaces $\mathcal{C}_d \subset \mathcal{C}$, where $d > 6$ with $d \equiv 0, 2 \bmod 6$. The hypersurface $\mathcal{C}_d$ parametrizes cubic fourfolds of discriminant d, that is, the set of cubic fourfolds X which contain a surface S such that the discriminant of the saturated lattice spanned by h_X^2 and $[S]$ in $H^{2,2}(X, \mathbb{Z}) := H^4(X, \mathbb{Z}) \cap H^2(\Omega_X^2)$ is d (here h_X stands for the class of a hyperplane section of X). For general results on cubic fourfolds, we refer the reader to [3; 4].

Another important case is that of smooth quadric hypersurfaces in a 5-dimensional linear section $\mathbb{V} \subset \mathbb{P}^8$ of the cone in $\mathbb{P}^{10}$ over the Grassmannian $\mathbb{G}(1, 4) \subset \mathbb{P}^9$. Such hypersurfaces in $\mathbb{V}$ are known as Gushel–Mukai fourfolds (GM fourfolds for short), and are parametrized by a moduli space $\mathcal{GM}$ of dimension 24. Outside a closed subset of codimension 2 in $\mathcal{GM}$, we have that the ambient fivefold $\mathbb{V}$ is smooth, and hence it is isomorphic to a hyperplane section of $\mathbb{G}(1, 4) \subset \mathbb{P}^9$. In such case, a GM fourfold $X \subset \mathbb{V}$ is called ordinary. The Noether–Lefschetz locus in $\mathcal{GM}$ is a countable union of hypersurfaces $\mathcal{GM}_d$, labeled by the possible values of the discriminant d, which are the integers $d > 8$ with $d \equiv 0, 2, 4 \bmod 8$. If $d \equiv 2 \bmod 8$, then $\mathcal{GM}_d$ is the union of two irreducible components $\mathcal{GM}'_d \cup \mathcal{GM}''_d$; otherwise it is irreducible. For general results on Gushel–Mukai fourfolds, we refer the reader to [1].

MSC2020: 14J35, 14J45, 68W30, 14Q10, 14-04.
Keywords: cubic fourfold, Gushel–Mukai fourfold, K3 surface.
SpecialFanoFourfolds version 2.7.1

One of the most important questions about cubic fourfolds and GM fourfolds is to establish when they are rational. There are classical examples of such fourfolds which are rational, but it is conjectured that most of them are not rational.

In the following section, we give just a brief introduction to how to use the package. For more computational details and recent applications of the package, see [12].

1. THE MAIN FUNCTIONS OF THE PACKAGE.

Constructing Hodge-special fourfolds. The first task of the package is to provide tools to create Hodge-special fourfolds and locate them in the corresponding Noether–Lefschetz locus. The most basic constructor of such objects is given by the function `specialFourfold`, which typically takes as input a pair (S, X) of a surface S and a fourfold X with $S \subset X$, and returns an object that can be used anywhere in place of X. It is possible to specify the ambient fivefold $\mathbb{V}$ as third argument. On the returned object we can apply the function `discriminant` to find the component of the Noether–Lefschetz locus where the fourfold belongs. Clearly, this depends on the surface S since the fourfold X could belong to several components.

In the following example, we take a random cubic fourfold $X \subset \mathbb{P}^5$ containing a plane S and verify that it belongs to the components $\mathcal{C}_8 \subset \mathcal{C}$. The value $d = 8$ comes from the fact that the self-intersection $(S)^2_X$ of S in X is 3, so that the discriminant of X is

$$\begin{aligned} d &= \det \begin{pmatrix} h_X^4 & h_X^2 \cdot S \\ S \cdot h_X^2 & (S)_X^2 \end{pmatrix} \\ &= \det \begin{pmatrix} \deg X & \deg S \\ \deg S & (S)_X^2 \end{pmatrix} \\ &= \det \begin{pmatrix} 3 & 1 \\ 1 & 3 \end{pmatrix} = 8, \end{aligned}$$

which is computed as

```
i1 : needsPackage "SpecialFanoFourfolds";
i2 : K = ZZ/65521;
i3 : S = random({3:{1}},0_(PP_K^5)); -- a random plane in PP^5
o3 : ProjectiveVariety, surface in PP^5
i4 : X = specialFourfold(S,random(3,S));
o4 : ProjectiveVariety, cubic fourfold containing a surface of degree 1 and sectional genus 0
i5 : discriminant X
o5 = 8
```

Note that in the line `i4`, we could omit the second argument by just typing `X = specialFourfold S`. In this case, the cubic fourfold X will be chosen randomly from those that contain S. The same applies in the case of GM fourfolds.

Below we construct an ordinary GM fourfold X as the transverse intersection of $\mathbb{G}(1,4)$ with a hyperplane and a hyperquadric through a plane $S \subset \mathbb{G}(1,4)$ of type $\sigma_{3,1}$. We verify that this fourfold belongs to the component $\mathcal{GM}''_{10}$.

```
i6 : S = schubertCycle({3,1},GG_K(1,4));

o6 : ProjectiveVariety, surface in PP^9 (subvariety of codimension 4 in GG(1,4) ⊂ PP^9)

i7 : X = specialFourfold S; -- random GM fourfold through S

o7 : ProjectiveVariety, GM fourfold containing a surface of degree 1 and sectional genus 0

i8 : discriminant X

o8 = 10

i9 : describe X

o9 = Special Gushel-Mukai fourfold of discriminant 10('')
     containing a surface in PP^8 of degree 1 and sectional genus 0
     cut out by 6 hypersurfaces of degree 1
     and with class in G(1,4) given by s_(3,1)
     Type: ordinary
```

Finally we construct a complete intersection of three quadrics in $\mathbb{P}^7$ containing a plane, thought of as a quadric hypersurface in a complete intersection of two quadrics $\mathbb{V} \subset \mathbb{P}^7$; see also [9]. We only specify the surface S, and let *Macaulay*2 choose the fourfold and the ambient fivefold $\mathbb{V}$ randomly.

```
i10 : S = random({5:{1}},0_(PP_K^7)); -- a random plane in PP^7

o10 : ProjectiveVariety, surface in PP^7

i11 : X = specialFourfold S;

o11 : ProjectiveVariety, complete intersection of three quadrics in PP^7
      containing a surface of degree 1 and sectional genus 0

i12 : discriminant X

o12 = 31
```

Count of parameters. A surface $S \subset \mathbb{V}$ corresponds to a point of some irreducible component $\mathcal{S}$ of the Hilbert scheme $\mathrm{Hilb}(\mathbb{V})$. Thus it is useful to have an estimate on the dimension of the family of fourfolds $X \subset \mathbb{V}$ containing some surface of $\mathcal{S}$. The function `parameterCount` applied to a fourfold $X \supset S$ automates a count of parameters based on the following proposition.

Proposition 1.1 ([7; 11; 12]; see also [5; 8]). *Let $S \subset \mathbb{V}$ be a smooth irreducible surface which is contained in a smooth hypersurface $X \subset \mathbb{V}$ of degree r. Assume that*

(1) $h^1(N_{S/\mathbb{V}}) = 0$, *and*

(2) $h^1(\mathcal{O}_S(r)) = 0$ *and* $h^0(\mathcal{I}_{S/\mathbb{V}}(r)) = h^0(\mathcal{O}_{\mathbb{V}}(r)) - \chi(\mathcal{O}_S(r))$.

Then there is a unique irreducible component $\mathcal{S} \subset \mathrm{Hilb}(\mathbb{V})$ of the Hilbert scheme of $\mathbb{V}$ that contains $[S]$, and the family $\mathfrak{X}_{\mathcal{S}} \subset \mathbb{P}(H^0(\mathcal{O}_{\mathbb{V}}(r)))$ of the hypersurfaces in $\mathbb{V}$ of degree r containing some surface of the family $\mathcal{S}$ has codimension at most

$$\dim\big(\mathbb{P}\big(H^0(\mathcal{O}_{\mathbb{V}}(r))\big)\big) - \big(h^0(N_{S/\mathbb{V}}) + h^0(\mathcal{I}_{S/\mathbb{V}}(r)) - h^0(N_{S/X}) - 1\big). \tag{1-1}$$

Particularly important is the case when the value of (1-1) is 1 since in this case we generally obtain a geometric description of a whole irreducible component of the Noether–Lefschetz locus. See [11] and [12] for a systematic use of the function `parameterCount` to deduces some birational properties of the first components of the Noether–Lefschetz locus in the moduli space of GM fourfolds; see also [7].

In the example below, we execute the function to deduce that the family of cubic fourfolds containing a plane is the whole component $\mathcal{C}_8$.

```
i13 : X = o4; -- cubic fourfold constructed in the input line i4

o13 : ProjectiveVariety, cubic fourfold containing a surface of
      degree 1 and sectional genus 0

i14 : parameterCount(X,Verbose=>true)
-- S: plane in PP^5
-- X: smooth cubic hypersurface in PP^5
-- h^1(N_{S,P^5}) = 0
-- h^0(N_{S,P^5}) = 9
-- h^1(O_S(3)) = 0, and h^0(I_{S,P^5}(3)) = 46 = h^0(O_(P^5)(3)) - \chi(O_S(3));
-- h^0(N_{S,X}) = 0
-- codim{[X] : S ⊂ X} <= 1

o14 = (1, (46, 9, 0))
```

Establishing rationality of fourfolds. The function `detectCongruence` may detect eventual congruences of $(re-1)$-secant curves of degree $e \geq 1$ to the surface S inside the ambient fivefold $\mathbb{V}$; see [8] for theoretical details. More precisely, it detects if for some $e \geq 1$ there exists a unique curve of degree e passing through the general point of $\mathbb{V}$, which is contained in $\mathbb{V}$ and intersects the surface S at $re-1$ points. In most cases, this is accomplished by considering the rational map $\varphi : \mathbb{V} \dashrightarrow Z \subset \mathbb{P}^N$ defined by the linear system of hypersurfaces of degree r in $\mathbb{V}$ through S. Indeed, if φ is birational onto its image $Z = \overline{\varphi(\mathbb{V})}$, it induces a 1–1 correspondence

$$\bigcup_{e\geq 1}\left\{\begin{array}{c}\text{curves of degree } e \text{ in } \mathbb{V} \text{ passing through a general}\\ \text{point } p\in\mathbb{V} \text{ and that are } (re-1)\text{-secant to } S\end{array}\right\} \xrightarrow{\simeq} \left\{\begin{array}{c}\text{lines contained in } Z\subset\mathbb{P}^N \text{ and}\\ \text{passing through } \varphi(p)\end{array}\right\}.$$

So that, one can analyze the pull-backs of the lines in $\mathbb{V}$ passing through a general point on Z; see the documentation for the functions `coneOfLines` and `RationalMap^* ProjectiveVariety`. The object returned looks like a rational map $h : \mathbb{V} \dashrightarrow \mathrm{Hilb}(\mathbb{V})$ which takes a point $p \in \mathbb{V}$ and returns the unique curve of degree e, $(re-1)$-secant to S, and passing through p.

In most interesting cases, the general curve of the congruence can be realized as the general fiber of the rational map $\mu : \mathbb{V} \dashrightarrow W = \overline{\mu(\mathbb{V})}$ defined by the linear system of hypersurfaces of degree $re-1$ in $\mathbb{V}$ having points of multiplicity e along the surface S (so that necessarily $\dim W = 4$). This map can be obtained with the command `map h`, where h is the congruence. The importance of the congruences is that, under a mild hypothesis of transversality (see [8]), we have that the restriction of the map μ induces a birational map $\mu|_X : X \dashrightarrow W$. So that X is rational if and only if W is.

In the example below, we show that a GM fourfold X containing a $\sigma_{3,1}$-plane

$$S \subset \mathbb{V} = \mathbb{P}^8 \cap \mathbb{G}(1,4) \subset \mathbb{P}^9$$

is rational. Indeed $S \subset \mathbb{V}$ admits a congruence of 1-secant lines from which we get a birational map from X to a smooth quadric hypersurface $W \subset \mathbb{P}^5$.

```
i15 : X = o7; -- GM fourfold constructed in the input line i7
o15 : ProjectiveVariety, GM fourfold containing a surface of degree 1 and sectional genus 0
i16 : (S, V) = (surface X, ambientFivefold X);
i17 : h = detectCongruence X;
o17 : Congruence of 1-secant lines to S in V
i18 : mu = map h;
o18 : RationalMap (dominant rational map from V to hypersurface in PP^5)
i19 : (mu|X)^-1;
o19 : RationalMap (birational map from hypersurface in PP^5 to X)
```

Computing associated K3 surfaces to rational fourfolds. Here let's assume that X is a cubic fourfold (resp., a GM fourfold) of discriminant d which contains a surface S that admits a congruence of curves, and moreover we can produce a birational map $\mu|_X : X \dashrightarrow W$ as described in the previous subsection. Then the inverse map of $\mu|_X$ is defined by a linear system of hypersurfaces in W with points of multiplicity e along a surface $U \subset W$, which turns out to be a projection of a K3 surface $\widetilde{U} \subset \mathbb{P}^g$ of degree d and genus $g = d/2 + 1$. For details and precise results, see [10]; see also [6] and [12]. The function `associatedK3surface` applied to the fourfold X returns this surface $\widetilde{U}$. As an example, we now compute the K3 surface $\widetilde{U} \subset \mathbb{P}^6$ associated with the GM fourfolds X containing a $\sigma_{3,1}$-plane.

```
i20 : U' = associatedK3surface X;
o20 : ProjectiveVariety, K3 surface associated to X
i21 : describe U'
o21 = ambient:.............. PP^6
      dim:.................. 2
      codim:................ 4
      degree:............... 10
      generators:........... 2^6
      purity:............... true
      dim sing. l.:......... -1
```

The function `associatedK3surface` can be used to construct explicit general K3 surfaces of given genus g. Currently, this can be done for $g \in \{6, 8, 11, 14, 20, 22\}$, and consequently for these values we get an explicit unirationality of the moduli space $\mathcal{F}_g$ of polarized K3 surfaces of genus g. See [12] and also [6; 10] for more explanations and references on this topic.

SUPPLEMENT. The online supplement contains version 2.7.1 of *SpecialFanoFourfolds*.

REFERENCES.

[1] O. Debarre, A. Iliev, and L. Manivel, "Special prime Fano fourfolds of degree 10 and index 2", pp. 123–155 in *Recent advances in algebraic geometry*, edited by C. D. Hacon et al., London Math. Soc. Lecture Note Ser. **417**, Cambridge Univ. Press, 2015. MR Zbl

[2] D. R. Grayson and M. E. Stillman, "Macaulay2, a software system for research in algebraic geometry", available at https://macaulay2.com/. Zbl

[3] B. Hassett, "Some rational cubic fourfolds", *J. Algebraic Geom.* **8**:1 (1999), 103–114. MR Zbl

[4] B. Hassett, “Special cubic fourfolds”, *Compositio Math.* **120**:1 (2000), 1–23. MR Zbl

[5] M. Hoff and G. Staglianò, “New examples of rational Gushel–Mukai fourfolds”, *Math. Z.* **296**:3–4 (2020), 1585–1591. MR Zbl

[6] M. Hoff and G. Staglianò, “Explicit constructions of $K3$ surfaces and unirational Noether–Lefschetz divisors”, *J. Algebra* **611** (2022), 630–650. MR Zbl

[7] H. Nuer, “Unirationality of moduli spaces of special cubic fourfolds and K3 surfaces”, *Algebr. Geom.* **4**:3 (2017), 281–289. MR Zbl

[8] F. Russo and G. Staglianò, “Congruences of 5-secant conics and the rationality of some admissible cubic fourfolds”, *Duke Math. J.* **168**:5 (2019), 849–865. MR Zbl

[9] F. Russo and G. Staglianò, “On complete intersections of three quadrics in $\mathbb{P}^7$”, preprint, 2023. Zbl arXiv 2312.01773

[10] F. Russo and G. Staglianò, “Trisecant flops, their associated K3 surfaces and the rationality of some cubic fourfolds”, *J. Eur. Math. Soc.* **25**:6 (2023), 2435–2482. MR Zbl

[11] G. Staglianò, “On some families of Gushel–Mukai fourfolds”, *Algebra Number Theory* **16**:8 (2022), 1905–1918. MR Zbl

[12] G. Staglianò, “Explicit computations with cubic fourfolds, Gushel–Mukai fourfolds, and their associated K3 surfaces”, *Exp. Math.* (online publication April 2023). Zbl

RECEIVED: 18 Aug 2022 REVISED: 18 Jan 2023 ACCEPTED: 14 Apr 2024

GIOVANNI STAGLIANÒ:

giovanni.stagliano@unict.it
Dipartimento di Matematica e Informatica, Università degli Studi di Catania, Catania, Italy